
Abstract. It is common to model inductive datatypes as least fixed
points of functors. We show that within the Cedille type theory we can
relax functoriality constraints and generically derive an induction prin-
ciple for Mendler-style lambda-encoded inductive datatypes, which arise
as least fixed points of covariant schemes where the morphism lifting is
defined only on identities. Additionally, we implement a destructor for
these lambda-encodings that runs in constant-time. As a result, we can
define lambda-encoded natural numbers with an induction principle and
a constant-time predecessor function so that the normal form of a nu-
meral requires only linear space. The paper also includes several more
advanced examples.

1 Introduction

It is widely known that inductive datatypes may be defined in pure impredicative
type theory. For example, Church encodings identify each natural number n

with its iterator λ s. λ z. sn z. The Church natural numbers can be typed in
System F by means of impredicative polymorphism:

cNat ◭ ⋆ = ∀ X : ⋆. (X → X) → X → X.

The first objection to lambda-encodings is that it is provably impossible to derive
an induction principle in second-order dependent type theory [1]. As a conse-
quence, most languages come with a built-in infrastructure for defining inductive
datatypes and their induction principles. Here are the definitions of natural num-
bers in Agda and Coq:

data Nat : Set Inductive nat : Type :=

zero : Nat | 0 : nat

suc : Nat → Nat | S : nat → nat.

http://arxiv.org/abs/1803.02473v1

Coq will automatically generate the induction principle for nat, and in Agda it
can be derived by pattern matching and explicit structural recursion.

Therefore, we can ask if it is possible to extend the Calculus of Constructions
with typing constructs that make induction derivable for lambda-encoded
datatypes. Stump gave a positive answer to this question by introducing the
Calculus of Dependent Lambda Eliminations (CDLE) [2]. CDLE is a Curry-
style Calculus of Constructions extended with implicit products, intersection
types, and primitive heterogeneous equality. Stump proved that natural number
induction is derivable in this system for lambda-encoded natural numbers. Later,
we generalized this work by deriving induction for lambda-encodings of inductive
datatypes which arise as least fixed points of functors [3]. Moreover, we observed
that the proof of induction for Mendler-style lambda-encoding relied only on the
identity law of functors. In this paper, we exploit this observation to define a
new class of covariant schemes, which includes functors, and induces a larger
class of inductive datatypes supporting derivable induction.

Another objection to lambda-encodings is their computational inefficiency.
For example, computing the predecessor of a Church encoded Peano natural
provably requires linear time [4]. The situation was improved by Parigot who
proposed a new lambda-encoding of numerals with a constant-time predecessor,
but the size of the number n is exponential O(2n) [5]. Later, the situation was
improved further by the Stump-Fu encoding, which supports a constant-time
predecessor and reduces the size of the natural number n to O(n2) [6]. In this
paper, we show how to develop a constant-time predecessor within CDLE for a
Mendler-style lambda-encoded naturals that are linear in space.

This paper makes the following technical contributions:

1. We introduce a new kind of parameterized scheme using identity mappings

(function lifting defined only on identities). We show that every functor has
an associated identity mapping, but not vice versa.

2. We use a Mendler-style lambda-encoding to prove that every scheme with an
identity mapping induces an inductive datatype. Additionally, we generically
derive an induction principle for these datatypes.

3. We implement a generic constant-time destructor of Mendler-style lambda-
encoded inductive datatypes. To the best of our knowledge, we offer a first
example of typed lambda-encoding of inductive datatypes with derivable
induction and a constant-time destructor where normal forms of data require
linear space.

4. We give several examples of concrete datatypes defined using our develop-
ment. We start by giving a detailed description of lambda-encoded naturals
with an induction principle and a constant-time predecessor function that
only requires linear space to encode a numeral. We also give examples of
infinitary datatypes. Finally, we present an inductive datatype that arises
as a least fixed point of a scheme that is not a functor, but has an identity
mapping.

2 Background

In this section, we briefly summarize the main features of Cedille’s type theory.
For full details on CDLE, including semantics and soundness results, please see
the previous papers [2,7]. The main metatheoretic property proved in the previ-
ous work is logical consistency: there are types which are not inhabited. CDLE
is an extrinsic (i.e., Curry-style) type theory, whose terms are exactly those of
the pure untyped lambda calculus (with no additional constants or constructs).
The type-assignment system for CDLE is not subject-directed, and thus cannot
be used directly as a typing algorithm. Indeed, since CDLE includes Curry-style
System F as a subsystem, type assignment is undecidable [8]. To obtain a usable
type theory, Cedille thus has a system of annotations for terms, where the anno-
tations contain sufficient information to type terms algorithmically. But true to
the extrinsic nature of the theory, these annotations play no computational role.
Indeed, they are erased both during compilation and before formal reasoning
about terms within the type theory, in particular by definitional equality (see
Figure 1).

Γ, x : T ′ ⊢ t : T x 6∈ FV(|t|)

Γ ⊢ Λx :T ′. t : ∀x :T ′. T

Γ ⊢ t : ∀ x :T ′. T Γ ⊢ t′ : T ′

Γ ⊢ t − t′ : [t′/x]T

Γ ⊢ t : T
Γ ⊢ β : t ≃ t

Γ ⊢ t′ : t1 ≃ t2 Γ ⊢ t : [t1/x]T

Γ ⊢ ρ t′ − t : [t2/x]T

Γ ⊢ t1 : T Γ ⊢ t2 : [t1/x]T
′ Γ ⊢ p : |t1| ≃ |t2|

Γ ⊢ [t1, t2{p}] : ι x :T. T
′

Γ ⊢ t : ι x :T. T ′

Γ ⊢ t.1 : T

Γ ⊢ t : ι x :T. T ′

Γ ⊢ t.2 : [t.1/x]T ′

|Λx :T. t| = |t|
|t − t′| = |t|
|β| = λx. x
|ρ t − t′| = |t′|
|[t1, t2{p}]| = |t1|
|t.1| = |t|
|t.2| = |t|

Fig. 1. Introduction, elimination, and erasure rules for additional type constructs

CDLE extends the (Curry-style) Calculus of Constructions (CC) with prim-
itive heterogeneous equality, intersection types, and implicit products:

– t1 ≃ t2, a heterogeneous equality type. The terms t1 and t2 are required to
be typed, but need not have the same type. We introduce this with a constant
β which erases to λ x. x (so our type-assignment system has no additional
constants, as promised); β proves t ≃ t for any typeable term t. Combined
with definitional equality, β proves t1 ≃ t2 for any β-equal t1 and t2 whose
free variables are all declared in the typing context. We eliminate the equality
type by rewriting, with a construct ρ t’ - t. Suppose t’ proves t1 ≃ t2
and we synthesize a type T for t, where T has several occurrences of terms
definitionally equal to t1. Then the type synthesized for ρ t’ - t is T except

with those occurrences replaced by t2. Note that the types of the terms are
not part of the equality type itself, nor does the elimination rule require
that the types of the left-hand and right-hand sides are the same to do an
elimination.

– ι x : T. T’, the dependent intersection type of Kopylov [9]. This is the type
for terms t which can be assigned both the type T and the type [t/x]T’, the
substitution instance of T’ by t. In the annotated language, we introduce
a value of ι x : T. T’ by construct [ t, t’ {p} ], where t has type T

(algorithmically), t’ has type [t/x]T’, and p proves t ≃ t’. There are
also annotated constructs t.1 and t.2 to select either the T or [t.1/x]T’
view of a term t of type ι x : T. T’.

– ∀ x : T. T’, the implicit product type of Miquel [10]. This can be thought
of as the type for functions which accept an erased input of type x : T, and
produce a result of type T’. There are term constructs Λ x. t for introducing
an implicit input x, and t -t’ for instantiating such an input with t’. The
implicit arguments exist just for purposes of typing so that they play no
computational role and equational reasoning happens on terms from which
the implicit arguments have been erased.

It is important to understand that the described constructs are erased before
the formal reasoning, according to the erasure rules in Figure 1.

We have implemented CDLE in a tool called Cedille, which we have used to
typecheck the developments of this paper. The pre-release version is here:

http://cs.uiowa.edu/~astump/cedille-prerelease.zip

The Cedille code accompanying this paper is here:

http://firsov.ee/efficient-lambda/itp2018-code.zip

3 Preliminaries

We skip the details of the lambda-encoded implementation of basic datatypes
like Unit, Empty, sums (X + Y), and dependent products (Σ x : X. Y x), for
which the usual introduction and elimination rules are derivable in Cedille.

In this paper, we use syntactical simplifications to improve readability. In
particular, we hide the type arguments in the cases when they are unambiguous.
For example, if x : X and y : Y then we wrtie pair x y instead of fully type-
annotated pair X Y x y. The current version of Cedille requires fully annotated
terms.

3.1 Multiple Types of Terms

CDLE’s dependent intersection types allow judgementally equal values to be
intersected. Given x : X, y : Y x, and a proof p of x ≃ y, we can introduce an
intersection value v := [ x, y {p} ] of type ι x : X. Y x. Every intersection
has two “views”: the first view v.1 has type X and the second view v.2 has
type Y x. The term [ x, y {p} ] erases to x according to the erasure rules in
Figure 1. This allows us to see x as having two distinct types, namely X and Y x:

http://cs.uiowa.edu/~astump/cedille-prerelease.zip
http://firsov.ee/efficient-lambda/itp2018-code.zip

subst ◭ ∀ X: ⋆. ∀ Y: X → ⋆. Π x: X. ∀ y: Y x. ∀ p: x ≃ y. Y x

= Λ X. Λ Y. λ x. Λ y. Λ p. [ x, y {p} ].2.

(Π x : X. T is usual “explicit” dependent function space;) Indeed, the definition
of subst erases to term λ x. x. Hence, subst x -y -p beta-reduces to x and
has type Y x (dash denotes the application of implicitly quantified arguments).

3.2 Identity Functions

In our setting, it is possible to implement a function of type X → Y so that it
erases to term λ x. x where X is different from Y. The simplest example is the
first (or second) “view” from an intersection value:

view1 ◭ ∀ X : ⋆. ∀ Y : X → ⋆. (ι x : X. Y x) → X

= Λ X. Λ Y. λ x. x.1.

Indeed, according to the erasure rules view1 erases to the term λ x. x. We
introduce a type Id X Y, which is the set of all functions from X to Y that erase
to the identity function (λ x. x):

id ◭ ∀ X : ⋆. X → X = Λ X. λ x. x.

Id ◭ ⋆ → ⋆ → ⋆ = λ X : ⋆. λ Y : ⋆. Σ f : X → Y. f ≃ id.

Introduction The importance of the previously implemented combinator subst
is that it allows to introduce an identity function Id X Y from any extensional
identity f : X → Y (i.e., f x ≃ x for any x):

intrId ◭ ∀ X Y : ⋆. Π f : X → Y. (Π x : X. f x ≃ x) → Id X Y

= Λ X. Λ Y. λ f. λ prf. pair (λ x. subst x -(f x) -(prf x)) β.

Elmination Given an identity function c : Id X Y and a value x : X we can
apply the identity function c to x so that elimId -c x has type Y:

elimId ◭ ∀ X Y : ⋆. ∀ c : Id X Y. X → Y =

= Λ X. Λ Y. Λ c. λ x. subst x -(π1 c x) -(ρ (π2 c) - β).

The subterm ρ (π2 c) - β proves π1 c x ≃ x, where πi is the i-th projec-
tions from a dependent product. Observe that elimId itself erases to λ x. x,
hence elimId -c x ≃ x by beta-reduction. In other words, an identity function
Id X Y allows x : X to be seen as having types X and Y at the same time.

3.3 Identity Mapping

A scheme F : ⋆ → ⋆ is a functor if it comes equipped with a function fmap

that satisfies the identity and composition laws:

Functor ◭ (⋆ → ⋆) → ⋆ = λ F : ⋆ → ⋆.

Σ fmap : ∀ X : ⋆. ∀ Y : ⋆. (X → Y) → F X → F Y.

IdentityLaw fmap × CompositionLaw fmap.

However, it is simple to define a covariant scheme for which the function fmap

cannot be implemented (below, x1 6= x2 is shorthand for x1 ≃ x2 → Empty):

UneqPair ◭ ⋆ → ⋆ = λ X : ⋆. Σ x1 : X. Σ x2 : X. x1 6= x2.

We introduce schemes with identity mappings as a new class of parameterized
covariant schemes. An identity mapping is a lifting of identity functions:

IdMapping ◭ (⋆ → ⋆) → ⋆ = λ F : ⋆ → ⋆.

∀ X Y : ⋆. Id X Y → Id (F X) (F Y).

Intuitively, IdMapping F is similar to fmap of functors, but it needs to be defined
only on identity functions. The identity law is expressed as a requirement that
identity function Id X Y is mapped to identity function Id (F X) (F Y).

Clearly, every functor induces an identity mapping (by the application of
intrId to fmap and its identity law):

fm2im ◭ ∀ F : ⋆ → ⋆. Functor F → IdMapping F = <..>

However, UneqPair is an example of scheme which is not a functor, but has an
identity mapping (see example in Section 6.3).

In the rest of the paper we show that every identity mapping IdMapping F

induces an inductive datatype which is a least fixed point of F. Additionally,
we generically derive an induction principle and implement a constant-time de-
structor for these datatypes.

4 Inductive Datatypes from Identity Mappings

In our previous paper, we used Cedille to show how to generically derive an
induction principle for Mendler-style lambda-encoded datatypes that arise as
least fixed points of functors [3]. In this section, we revisit this derivation to
show that it is possible to relax functoriality constraints and only assume that
the underlying signature scheme is accompanied by an identity mapping.

4.1 Basics of Mendler-Style Encoding

In this section, we investigate the standard definitions of Mendler-style F-algebras
that are well-defined for any unrestricted scheme F : ⋆ → ⋆. To reduce the no-
tational clutter, we assume that F : ⋆ → ⋆ is a global (module) parameter:

module _ (F : ⋆ → ⋆)

In the abstract setting of category theory, a Mendler-style F-algebra is a pair
(X,Φ) where X is an object (i.e., the carrier) in C and Φ : C(−, X) → C(F −, X)
is a natural transformation. In the concrete setting of Cedille, objects are types,
arrows are functions, and natural transformations are polymorphic functions.
Therefore, Mendler-style F-algebras are defined as follows:

AlgM ◭ ⋆ → ⋆ = λ X : ⋆. ∀ R : ⋆. (R → X) → F R → X.

Uustalu and Vene showed that initial Mendler-style F-algebras offer an alter-
native categorical model of inductive datatypes [11]. The carrier of an initial
F-algebra is an inductive datatype that is a least fixed point of F. It is known
that if F is a positive scheme then the least fixed point of it may be implemented
in terms of universal quantification [12]:

FixM ◭ ⋆ = ∀ X : ⋆. AlgM X → X.

foldM ◭ ∀ X : ⋆. AlgM X → FixM → X = Λ X. λ alg. λ x. x alg.

In essence, this definition identifies inductive datatypes with iterators and every
function on FixM is to be computed by iteration.

The natural transformation of the initial Mendler-style F-algebra denotes the
collection of constructors of its carrier [11]. In our setting, the initial Mendler-
style F-algebra AlgM FixM is not defineable because F is not a functor [3]. Instead,
we express the collection of constructors of datatype FixM as a conventional F-
algebra F FixM → FixM:

inFixM ◭ F FixM → FixM = λ x. Λ X. λ alg. alg (fold alg) x.

The function inFixM is of crucial importance because it expresses constructors
of FixM without requirements of functoriality on F : ⋆ → ⋆.

It is provably impossible to define the mutual inverse of inFixM (destructor
of FixM) without introducing additional constraints on F. Assume the existence
of function outFixM (which need not be an inverse of inFixM), typed as follows:

outFixM ◭ ∀ F : ⋆ → ⋆. FixM F → F (FixM F) = <..>

Next, recall that in the impredicative setting the empty type is encoded as
∀ X : ⋆. X (its inhabitant implies any equation). Then, we instantiate F with
the negative polymorphic scheme NegF X := ∀ Y: ⋆. X → Y, and exploit the
function outFixM to construct a witness of the empty type:

T ◭ ⋆ = FixM NegF.

ty ◭ ∀ Y : ⋆. T → Y = Λ Y. λ t. outFixM NegF t Y t.

t ◭ T = ty T (inFixM NegF ty).

unsound ◭ ∀ X : ⋆. X = Λ X. ty X t.

Therefore, the existance of function outFixM contradicts the consistency of
Cedille. Hence, the inverse of inFixM can exist only for some restricted class
of schemes F : ⋆ → ⋆.

4.2 Inductive Subset

From this point forward we assume that the scheme F is also accompanied by
an identity mapping imap:

module _ (F : ⋆ → ⋆)(imap : IdMapping F)

In our previous work we assumed that F is a functor and showed how to
specify the “inductive” subset of the type FixM F. Then, we generically derived
induction for this subset. In this section, we update the steps of our previous
work to account for F : ⋆ → ⋆ not being a functor.

The dependent intersection type ι x : X. Y x can be understood as a sub-
set of X defined by a predicate Y. However, to construct the value of this type
we must provide x : X and a proof p : Y x so that x and p are provably equal
(x ≃ p). Hence, to align with this constraint we use implicit products to ex-
press inductivity of FixM as its “dependently-typed” version. Recall that FixM

is defined in terms of Mendler-style F-algebras:

AlgM ◭ ⋆ → ⋆ = λ X : ⋆. ∀ R : ⋆. (R → X) → F R → X.

In our previous work, we introduced the Q-proof F-algebras as a “dependently-
typed” counterpart of AlgM. The value of type PrfAlgM X Q alg should be un-
derstood as an inductive proof that predicate Q holds for every X where X is a
least fixed point of F and alg : F X → X is a collection of constructors of X.

PrfAlgM ◭ Π X : ⋆. (X → ⋆) → (F X → X) → ⋆

= λ X : ⋆. λ Q : X → ⋆. λ alg : F X → X.

∀ R : ⋆. ∀ c : Id R X. (Π r : R. Q (elimId -c r)) →
Π fr : F R. Q (alg (elimId -(imap c) fr)).

Mendler-style F-algebras (AlgM) allow recursive calls to be explicitly stated by
providing arguments R → X and F R, where the polymorphically quantified type
R ensures termination. Similarly, Q-proof F-algebras allow the inductive hypothe-
ses to be explicitly stated for every R by providing an implicit identity function
c : Id R X, and a dependent function of type Π r : R. Q (elimId -c r) (re-
call that elimId -c r reduces to r and has type X). Given the inductive hypoth-
esis for every R, the proof algebra must conclude that the predicate Q holds for
every X, which is produced by constructors alg from any given F R that has
been “casted” to F X.

Next, recall that FixM is defined as a function from AlgM X to X for every X.

FixM ◭ ⋆ = ∀ X : ⋆. AlgM X → X.

To retain the analogy of definitions, we express the inductivity of value x : FixM

as a dependent function from a Q -proof F -algebra to Q x.

IsIndFixM ◭ FixM → ⋆ = λ x : FixM.

∀ Q : FixM → ⋆. PrfAlgM FixM Q inFixM → Q x.

Now, we employ intersection types to define a type FixIndM as a subset of FixM
carved out by the “inductivity” predicate IsIndFixM:

FixIndM ◭ ⋆ = ι x : FixM. IsIndFixM x.

Finally, we must explain how to construct the values of this type. As in the case of
FixM, the set of constructors of FixIndM is expressed by a conventional F-algebra
F FixIndM → FixIndM. The implementation is divided into three steps:

First, we define a function from F FixIndM to FixM:

tc1 ◭ F FixIndM → FixM = λ x.

let c ◭ Id (F FixIndM) (F FixM) = imap (intrId (λ x. x.1) β) in

inFixM (elimId -c x).

The implementation simply “casts” its argument to F FixM and then applies
the previously implemented constructor of FixM (inFixM). Because elimId -c x

reduces to x, the erasure of tc1 is the same as the erasure of inFixM which is a
term λ x. λ q. q (λ r. r q) x.

Second, we show that the same lambda term could also be typed as a proof
that every tc1 x is inductive:

tc2 ◭ Π x : F FixIndM. IsIndFixM (tc1 x)

= λ x. (Λ Q. λ q. (q -(intrId (λ x. x.1) β) (λ r. r.2 q) x)).

Indeed, functions tc1 and tc2 are represented by the same pure lambda term.
Finally, given any value x : F FixInd we can intersect tc1 x and the proof

of its inductivity tc2 x to construct an element of an inductive subset FixIndM:

inFixIndM ◭ F FixIndM → FixIndM = λ x. [ tc1 x, tc2 x { β } ].

Recall that erasure of intersection [ x, y {p} ] equals the erasure of x. There-
fore, functions tc1, tc2, and inFixIndM all erase to the same pure lambda term.
In other words, in Cedille the term λ x. λ q. q (λ r. r q) x can be extrin-
sically typed as any of these functions.

4.3 Induction Principle

We start by explaining why we need to derive induction for FixIndM, even though
it is definitionally an inductive subset of FixM. Indeed, every value x : FixIndM

can be “viewed” as a proof of its own inductivity. More precisely, the term x.2

is a proof of the inductivity of x.1. Moreover, the equational theory of CDLE
gives us the following equalities x.1 ≃ x ≃ x.2 (due to the rules of erasure).
But recall that the inductivity proof provided by the second view x.2 is typed
as follows:

∀ Q : Fix → ⋆. PrfAlgM FixM Q inFixM → Q x.1

Note that Q is a predicate on FixM and not FixIndM! This form of inductivity
does not allow properties specified directly for FixIndM to be proven.

Therefore, our goal is to prove that every x : FixIndM is inductive in its
own right. We phrase this in terms of proof-algebras parameterized by FixIndM,
a predicate on FixIndM, and its constructors (inFixIndM):

∀ Q : FixIndM → ⋆. PrfAlgM FixIndM Q inFixIndM → Q x.

In our previous work, we already made an observation that the derivation of in-
duction for Mendler-style encodings relies only on the identity law of functors [3].
Therefore, the current setting only requires minor adjustments of our previous
proof. For the sake of completeness, we present a main idea of this derivation.

The key insight is that we can convert predicates on FixIndM to logically
equivalent predicates on FixM by using heterogeneous equality:

Lift ◭ (FixInd → ⋆) → Fix → ⋆

= λ Q : FixInd → ⋆. λ y : Fix. Σ x : FixInd. x ≃ y × Q x.

eqv1 ◭ Π x: FixIndM. ∀ Q: FixIndM → ⋆. Q x → Lift Q x.1 = <..>

eqv2 ◭ Π x: FixIndM. ∀ Q: FixIndM → ⋆. Lift Q x.1 → Q x = <..>

These properties allow us to convert a Q -proof algebra to a proof algebra for a
lifted predicate Lift Q, and then derive the generic induction principle:

convIH ◭ ∀ Q : FixIndM → ⋆. PrfAlgM FixIndM Q inFixIndM

→ PrfAlgM FixM (Lift Q) inFixM = <..>

induction ◭ ∀ Q: FixIndM → ⋆. PrfAlgM FixIndM Q inFixIndM →
Π e: FixIndM. Q e = Λ Q. λ p. λ e. eqv2 e (e.2 (convIH p)).

Let Q be a predicate on FixIndM and p be a Q -proof algebra: we show that Q
holds for any e : FixIndM. Recall that every e : FixIndM can be viewed as
a proof of inductivity of e.1 via e.2 : IsIndFixM e.1. We use this to get a
proof of the lifted predicate Lift Q e.1 from the proof algebra delivered by
convIH p. Finally, we get Q e by using eqv2.

5 Constant-Time Destructors

An induction principle is needed to prove properties about programs, but prac-
tical functional programming also requires constant-time destructors (also called
accessors) of inductive datatypes. Let us illustrate the problem using the datatype
of natural numbers. In Agda it is easy to implement the predecessor function by
pattern matching:

pred : Nat → Nat

pred zero = zero

pred (suc n) = n

The correctness of pred trivially follows by beta-reduction:

predProp : (n : Nat) → pred (suc n) ≡ n

predProp n = refl

Let us switch to Cedille and observe that it is much less trivial to implement
the predecessor for the impredicative encoding of Peano numerals. Here is the
definition of Church encoded Peano naturals and their constructors:

cNat ◭ ⋆ = ∀ X : ⋆. (X → X) → X → X.

zero ◭ cNat = Λ X. λ s. λ z. z.

suc ◭ cNat → cNat = λ n. Λ X. λ s. λ z. s (n s z).

Next, we implement the predecessor for cNat which is due to Kleene:

zCase ◭ cNat × cNat = pair zero zero

sCase ◭ cNat × cNat → cNat × cNat = λ n. pair (π2 n) (suc (π2 n)).

predK ◭ Nat → Nat = λ n. π1 (n sC zC).

The key to the Kleene predecessor is the function sCase, which ignores the first
item of the input pair, moves the second natural to the first position, and then
applies the successor of the second element within the second position. Hence,
folding a natural number n with sCase and zCase produces a pair (n-1, n). In
the end, predK n projects the first element of a pair.

Kleene predecessor runs in linear time. Also, predK (suc n) gets stuck after
reducing to π1 (pair (π2 (n sCase zCase))(suc (π2 (n sCase zCase)))).
Hence, we must use induction to prove that predK (suc n) computes to n.

Furthermore, Parigot proved that any definition of predecessor for the Church-
style lambda-encoded numerals requires linear time [4].

5.1 Constant-Time Destructor for Mendler-Style Encoding

In previous sections we defined a datatype FixIndM for every scheme F that has
an identity mapping. Then, we implemented the constructors of the datatype
as the function inFixIndM, and defined an induction principle phrased in terms
of this function. In this section, we develop a mutual inverse of inFixIndM that
runs in constant time. As a simple consequence, we prove that FixIndM is a least
fixed point of F.

Let us start by exploring the computational behaviour of the function foldM.
The following property is a variation of the cancellation law for Mendler-style
encoded data [11], and its proof is simply by beta-reduction.

foldHom ◭ ∀ X : ⋆. Π x : F FixM. Π alg : AlgM X.

foldM alg (inFixM x) ≃ alg (foldM alg) x = Λ X. λ x. λ a. β.

In other words, folding the inductive value inFixM x replaces its outermost
“constructor” inFixM with partially applied F-algebra alg (foldM alg).

It is well-known that (computationally) induction can be reduced to iteration
(folding). Therefore, we can state the cancellation law for the induction rule in
terms of proof algebras.

indHom ◭ ∀ Q : FixIndM → ⋆. Π alg : PrfAlgM FixIndM Q inFixIndM.

Π x : F FixIndM. Π c : Id FixIndM FixIndM.

induction alg (inFixInd x) ≃ alg -c (induction alg) x

= Λ Q. λ p. λ x. β.

Most importantly, is that the proof of indHom is by reflexivity (β), which en-
sures that the left-hand side of equality beta-reduces to the right-hand side in a
constant number of beta-reductions.

Next, we implement a proof algebra for the constant predicate λ _. F FixIndM.

outAlgM ◭ PrfAlgM FixIndM (λ _. F FixIndM) inFixIndM

= Λ R. Λ c. λ f. λ y. elimId -(imap c) y.

The identity mapping of F lifts the identity function c : Id R X to an iden-
tity function Id (F R) (F FixIndM), which is then applied to the argument
y : F R to get the desired value of F FixIndM.

The proof algebra outAlgM induces the constant-time inverse of inFixIndM:

outFixIndM ◭ FixInd → F FixInd = induction outAlgM.

Definitionally, outFixIndM (inFixIndM x) is induction outAlgM (inFixInd x),
which reduces to outAlgM -c (induction outAlgM) x in a constant number
of steps (indHom). Because outAlgM -c erases to λ f. λ y. y, it follows that
outFixIndM computes an inverse of inFixIndM in a constant number of beta-
reductions:

lambek1 ◭ Π x: F FixInd. outFixIndM (inFixIndM x) ≃ x = λ x. β.

Furthermore, we show that outFixIndM is a post-inverse:

lambek2 ◭ Π x: FixIndM. inFixIndM (outFixIndM x) ≃ x

= λ x. induction (Λ R. Λ c. λ ih. λ fr. β) x.

This direction requires us to invoke induction to “pattern match” on the argu-
ment value to get x := inFixIndM y for some value y of type F FixIndM. Then,
inFixIndM (outFixIndM (inFixIndM y)) ≃ inFixIndM y because the inner
term outFixIndM (inFixIndM y) is just y by beta reduction (lambek1).

6 Examples

In this section, we demonstrate the utility of our derivations on three examples.
First, we present a detailed implementation of natural numbers with a constant-
time predecessor function. Second, we show examples of infinitary datatypes.
Finally, we give an example of a datatype arising as a least fixed point of a
scheme that is not a functor, but has an identity mapping.

6.1 Natural Numbers with Constant-Time Predecessor

Natural numbers arise as a least fixed point of the functor NF:

NF ◭ ⋆ → ⋆ = λ X : ⋆. Unit + X.

nfmap ◭ Functor NF = <..>

Since every functor has an identity mapping then we use our framework to define
natural numbers as shown below:

nfimap ◭ IdMapping NF = fm2im nfmap.

Nat ◭ ⋆ = FixInd NF nfimap.

zero ◭ Nat = inFixIndM (in1 unit).

suc ◭ Nat → Nat = λ n. inFixIndM (in2 n).

If injections in1 and in2 erase to λ a. λ i. λ j. i a and λ a. λ i. λ j. j a,
respectively, then the natural number constructors have the following erasures:

zero ≃ λ alg. (alg (λ f. (f alg)) (λ i. λ j. (i (λ x. x))))

suc n ≃ λ alg. (alg (λ f. (f alg)) (λ i. λ j. (j n)))

Intuitively, Mendler-style numerals have a constant-time predecessor because
every natural number suc n contains the previous natural n as its direct subpart
(which is not true for Church encoding).

We implement the predecessor for Nat in terms of the generic constant-time
destructor outFixIndM:

pred ◭ Nat → Nat = λ n. case (outFixIndM n) (λ _. zero) (λ m. m).

Because elimination of disjoint sums (via case) and outFixIndM are both constant-
time operations, pred is also a constant-time function and its correctness is
immediate (i.e., by beta-reduction):

predSuc ◭ Π n : Nat. pred (suc n) ≃ n = λ n. β.

We also show that the usual “flat” induction principle can be derived from
our generic induction principle (induction) by dependent elimination of NF:

indNat ◭ ∀ P : Nat → ⋆. (Π n : Nat. P n → P (suc n)) → P zero

→ Π n : Nat. P n = Λ P. λ s. λ z. λ n. induction P

(Λ R. Λ c. λ ih. λ v. case v (λ u. ρ (etaUnit u) - z)

(λ r. s (elimId -c r) (ih r))) n.

6.2 Infinitary Trees

In Agda, we can give the following inductive definition of infinitary trees:

data ITree : Set where

node : (Nat → Unit + ITree) → ITree

ITree is a least fixed point of functor IF X := Nat → Unit + X. In Cedille,
we can implement a functorial function lifting for IF:

itfmap ◭ ∀ X Y : ⋆. (X → Y) → IF X → IF Y

= λ f. λ t. λ n. case (t n) (λ u. in1 u) (λ x. in2 (f x)).

To our best knowledge, it is impossible to prove that itfmap satisfies the func-
torial laws without functional extensionality (which is unavailable in Cedille).
However, it is possible to implement an identity mapping for the scheme IF:

itimap ◭ IdMapping IF

= Λ c. pair (λ x. λ n. elimId -(nfimap -c) (x n)) β.

The first element of a pair erases to λ x. λ n. x n, which is λ x. x by the eta
law1. Now, since we showed that IF has an identity mapping then our generic
development induces the datatype ITree with its constructor, destructor, and
induction principle.

ITree ◭ ⋆ = FixIndM IF itimap.

inode ◭ (Nat → Unit + ITree) → ITree = λ f. inFixIndM f.

The specialized induction is phrased in terms of “empty tree” iempty which acts
as a base case (projR “projects” a tree from disjoint sum or returns iempty):

iempty ◭ ITree = inode (λ _. in1 unit).

indITree ◭ ∀ P : ITree → ⋆. P iempty →
(Π f: Nat → Unit + ITree. (Π n : Nat. P (projR (f n))

→ P (inode f)) → Π t: ITree. P t = <..>

Next, let us look at another variant of infinitary datatypes in Agda:

data PTree : Set where

pnode : ((PTree → Bool) → Unit + PTree) → PTree

This definition will be rejected by Agda (and Coq) since it arises as a least fixed
point of the scheme PF X := Unit + ((X → Bool) → X) → X, which is pos-
itive but not strictly positive. The definition is rejected because it is currently
unclear if non-strict definitions are sound in Agda. For the Coq setting, there
is a proof by Coquand and Paulin that non-strict positivity combined with an
impredicative universe and a predicative universe hieararchy leads to inconsis-
tency [14]. In Cedille, we can implement an identity mapping for the scheme PF
in a similar fashion as the previously discussed UF. Hence, the datatype induced
by PF exists in the type theory of Cedille.

6.3 Unbalanced Trees

Consider the following definition of “unbalanced” binary trees in Agda:

1 Our analysis of CDLE up to now has included only β-equality. It is known that η
can cause problems for intrinsic type theories due to non-confluence of βη-reduction
on ill-typed terms (cf. [13]). But for extrinsic typing, we can use confluence of βη-
reduction on pure lambda terms, and thus we hope that adding η does cause prob-
lems. The use of η is confined to the example of infinitary trees only.

data UTree : Set where

leaf : Bool → UTree

node : (b1 : UTree) → (b2 : UTree) → b1 6= b2 → UTree

The datatype UTree arises as a least fixed point of the following scheme:

UF ◭ ⋆ → ⋆ = λ X : ⋆. Bool + (Σ x1 : X. Σ x2 : X. x1 6= x2).

Because the elements x1 and x2 must be different, lifting an arbitrary function
X → Y to UF X → UF Y is impossible. Hence, the scheme UF is not a functor.

However, we can show that UF has an identity mapping. We start by produc-
ing a function UF X → UF Y from an identity Id X Y:

uimap’ ◭ ∀ X Y : ⋆. ∀ i : Id X Y. UF X → UF Y = Λ i. λ uf.

case uf (λ u. in1 u)

(λ u. in2 (pair (elimId -i (π1 u))

(pair (elimId -i (π1 (π2 u))) (π2 (π2 u)))))

We prove that uimap’ -i is extensionally an identity function:

uimP ◭ ∀ X Y: ⋆. ∀ i: Id X Y. Π u: UF X. uimap’ -i u ≃ u = <..>

This is enough to derive an identity mapping for UF by using the previously
implemented combinator intrId:

uimap ◭ IdMapping UF = intrId uimap’ uimP.

Therefore, we conclude that the datatype of unbalanced trees exists in Cedille
and can be defined as a least fixed point of scheme UF:

UTree ◭ ⋆ = FixIndM UF uimap.

The specialized constructors, induction principle, and a destructor function for
UTree are easily derived from their generic counterparts (inFixIndM, induction,
outFixIndM).

7 Related Work

Pfenning and Paulin-Mohring show how to model inductive datatypes using
impredicative encodings in the Calculus of Constructions (CC) [15]. Because
induction is not provable in the CC, the induction principles are generated and
added as axioms. This approach was adopted by initial versions of the Coq proof
assistant, but later Coq switched to the Calculus of Inductive Constructions
(CIC), which has built-in inductive datatypes.

Delaware et al. derived induction for impredicative lambda-encodings in Coq
as a part of their framework for modular definitions and proofs (using the à la
carte technique [16]). They showed that a value v : Fix F is inductive if it is
accompanied by a proof of the universal property of folds [17].

Similarly, Torrini introduced the predicatisation technique, reducing depen-
dent induction to proofs that only rely on non-dependent Mendler induction (by
requiring the inductive argument to satisfy extra predicatisation hypothesis) [18].

Traytel et al. present a framework for constructing (co)datatypes in HOL [19,20].
The main ingredient is a notion of a bounded natural functor (BNF), or a bi-
nary functor with additional structure. BNFs are closed under composition and
fixed points, which enables support for both mutual and nested (co)recursion
with mixed combinations of datatypes and codatatypes. The authors developed
a package that can generate (co)datatypes with their associated proof-principles
from user specifications (including custom bounded natural functors). In con-
strast, our approach provides a single generic derivation of induction within the
theory of Cedille, but does not address codatatypes. It would be interesting to
further investigate the exact relationship between schemes with identity map-
pings and BNFs.

Church encodings are typeable in System F and represent datatypes as their
own iterators. Parigot proved that the lower bound of the predecessor function
for Church numerals has linear time complexity [4].

Parigot designed an impredicative lambda-encoding that is typeable in Sys-
tem Fω with positive-recursive type definitions. The encoding identifies datatypes
with their own recursors, allowing constant time destructors to be defined, but
the drawback is that the representation of a natural number n is exponential in
the call-by-value setting [5].

The Stump-Fu encoding is also typeable in System Fω with positive-recursive
type definitions. It improves upon the Parigot representation by requiring only
quadratic space, and it also supports constant-time destructors [6].

8 Conclusions and Future Work

In this work, we showed that the Calculus of Dependent Lambda Eliminations is
a compact pure type theory that allows a general class of Mendler-style lambda-
encoded inductive datatypes to be defined as least fixed points of schemes with
identity mappings. We also gave a generic derivation of induction and imple-
mented a constant-time destructor for these datatypes. We used our develop-
ment to give the first example (to the best of our knowledge) of lambda-encoded
natural numbers with: provable induction, a constant-time predecessor function,
and a linear size (in the numeral n) term representation. Our formal development
is around 700 lines of Cedille code.

For future work, we plan to explore coinductive definitions and to use the
categorical model of Mendler-style datatypes to investigate histomorphisms and
inductive-recursive datatypes in Cedille [11].

